

Structural Equation Modeling with Latent Variables or their and Plausible Values using MLR Mplus 7.11

These data were adapted from my dissertation work in which 152 adults age 63–87 years were measured on visual impairment (distance acuity and five degrees of contrast sensitivity), processing speed, divided visual attention, and selective visual attention (as measured by the Useful Field of View subtests for each), attentional search efficiency (DriverScan), and simulator driving impairment (as measured by six driving performance indicators).

Hoffman, L., Yang, X., Bovaird, J. A., & Embretson, S. E. (2006). Measuring attention in older adults: Development and psychometric evaluation of DriverScan. *Educational and Psychological Measurement*, 66, 984-1000.

Hoffman, L., McDowd, J. M., Atchley, P., & Dubinsky R. A. (2005). The role of visual attention in predicting driving impairment in older adults. *Psychology and Aging*, 20(4), 610-622.

This example will demonstrate how to estimate structural equation models, including models with latent variable interactions. But because simultaneous estimation of all effects of interest may not always be possible, this example will also how to generate, merge, and use plausible values instead. Finally, the example will also demonstrate the optimism of the results when using only a single factor score per person.

Mplus Code to Read in Data:

```

TITLE:      SEM Example for Driverscan
DATA:      FILE = driverscanSEM.dat;      ! FILE is file to be analyzed
              FORMAT IS free;                ! Free is default
              TYPE IS INDIVIDUAL;           ! Individual data is default

VARIABLE:  ! Every variable in data set
              NAMES = PartID sex age75 lncs15 lncs3 lncs6 lncs12 lncs18 far lnp
                lnda lnsa Dscan lane da_task crash stop speed time ticket acc;
              ! Every variable in THIS MODEL
              USEVARIABLES = lncs15 lncs3 lncs6 lncs12 lncs18 far;
              IDVARIABLE IS PartID;          ! Will keep ID variable for merging
              MISSING ARE ALL (-9999);       ! Make sure to specify all missing values

ANALYSIS:  ESTIMATOR IS MLR; ! For continuous items whose residuals may not be normal

OUTPUT:    SAMPSTAT                    ! Sample descriptives to verify data
              MODINDICES (4)              ! Voodoo to improve model (at p<.05)
              STDYX                        ! Requests fully standardized solution
              RESIDUAL                      ! Requests standardized and normalized residuals
              SVALUES;                      ! Write code with estimated parameters
              TECH4;                        ! Latent variable correlation matrix

SAVEDATA:  SAVE = FSCORES; FILE IS FactorScores.dat; ! Change .dat name by model

MODEL:    (model syntax goes here, to be changed for each model)

```

We will begin by fitting single-factor measurement models for each latent factor. This is for 2 reasons: (1) we need to ensure each factor fits *per se*, and (2) we will generate the plausible values to use later.

Measurement Model for Visual Impairment (including Omega)**Step 1: Estimate model using MLR, generate SVALUES text**

```

VARIABLE:  ! Every variable in THIS MODEL
           USEVARIABLES = lncs15 lncs3 lncs6 lncs12 lncs18 far;
ANALYSIS:  ESTIMATOR = MLR;
MODEL:     ! Measurement models
  Vision BY far@1 ! 1 marker loading
           lncs15* lncs3* lncs6* lncs12* lncs18* (L2-L6);
  [far* lncs15* lncs3* lncs6* lncs12* lncs18*];           ! All intercepts
  far* lncs15* lncs3* lncs6* lncs12* lncs18* (E1-E6);     ! Residual variances
  Vision*; [Vision@0];                                     ! Factor M=0, Var=?

```

```

MODEL CONSTRAINT: ! TO GET OMEGA
NEW(SumLoad2 SumError SumRCov Omega);
SumLoad2 = (1+L2+L3+L4+L5+L6)^2;
SumError = E1+E2+E3+E4+E5+E6;
SumRCov = 2*(0);
Omega = SumLoad2 / (SumLoad2+SumError+SumRCov);

```

MODEL FIT INFORMATION

Number of Free Parameters	18
---------------------------	----

Loglikelihood

H0 Value	-747.948
H0 Scaling Correction Factor for MLR	1.1255
H1 Value	-739.282
H1 Scaling Correction Factor for MLR	1.1171

Information Criteria

Akaike (AIC)	1531.897
Bayesian (BIC)	1586.327
Sample-Size Adjusted BIC	1529.357
(n* = (n + 2) / 24)	

Chi-Square Test of Model Fit

Value	15.752*
Degrees of Freedom	9
P-Value	0.0722
Scaling Correction Factor for MLR	1.1003

* The chi-square value for MLM, MLMV, MLR, ULSMV, WLSM and WLSMV cannot be used for chi-square difference testing in the regular way. MLM, MLR and WLSM chi-square difference testing is described on the Mplus website. MLMV, WLSMV, and ULSMV difference testing is done using the DIFFTEST option.

RMSEA (Root Mean Square Error Of Approximation)

Estimate	0.070
90 Percent C.I.	0.000 0.126
Probability RMSEA <= .05	0.246

CFI/TLI

CFI	0.973
TLI	0.955

Chi-Square Test of Model Fit for the Baseline Model

Value	264.950
Degrees of Freedom	15
P-Value	0.0000

SRMR (Standardized Root Mean Square Residual)

Value	0.041
-------	-------

Measurement Model Results from Step 1 for Vision:

MODEL RESULTS

VISION BY	Estimate	S.E.	Est./S.E.	Two-Tailed P-Value
FAR	1.000	0.000	999.000	999.000
LNCS15	0.497	0.103	4.815	0.000
LNCS3	0.594	0.118	5.018	0.000
LNCS6	0.764	0.136	5.628	0.000
LNCS12	1.296	0.207	6.277	0.000
LNCS18	1.504	0.237	6.353	0.000
Means				
VISION	0.000	0.000	999.000	999.000
Intercepts				
LNCS15	-3.698	0.035	-105.136	0.000
LNCS3	-3.938	0.035	-113.273	0.000
LNCS6	-3.730	0.043	-87.639	0.000
LNCS12	-2.368	0.066	-36.000	0.000
LNCS18	-1.406	0.081	-17.389	0.000
FAR	3.026	0.067	45.130	0.000
Variances				
VISION	0.224	0.067	3.333	0.001
Residual Variances				
LNCS15	0.133	0.018	7.435	0.000
LNCS3	0.105	0.014	7.451	0.000
LNCS6	0.145	0.028	5.231	0.000
LNCS12	0.282	0.047	5.947	0.000
LNCS18	0.488	0.062	7.933	0.000
FAR	0.460	0.055	8.349	0.000
New/Additional Parameters				
SUMLOAD2	31.983	7.564	4.228	0.000
SUMERROR	1.613	0.102	15.822	0.000
SUMRCOV	0.000	0.000	0.000	1.000
OMEGA	0.952	0.011	88.539	0.000

The SVALUES output option provided this code, which we will use to generate plausible values for our factor scores for later.

```
MODEL COMMAND WITH FINAL ESTIMATES USED AS STARTING VALUES

vision BY far@1;
vision BY lnCS15*0.49665 (12);
vision BY lnCS3*0.59433 (13);
vision BY lnCS6*0.76361 (14);
vision BY lnCS12*1.29636 (15);
vision BY lnCS18*1.50436 (16);

[ lnCS15*-3.69842 ];
[ lnCS3*-3.93821 ];
[ lnCS6*-3.72997 ];
[ lnCS12*-2.36777 ];
[ lnCS18*-1.40608 ];
[ far*3.02632 ];
[ vision@0 ];

lnCS15*0.13297 (e2);
lnCS3*0.10479 (e3);
lnCS6*0.14501 (e4);
lnCS12*0.28191 (e5);
lnCS18*0.48808 (e6);
far*0.46001 (e1);
vision*0.22350;
```

STANDARDIZED MODEL RESULTS

STDYX Standardization

VISION BY	Estimate	S.E.	Est./S.E.	Two-Tailed P-Value
FAR	0.572	0.062	9.190	0.000
LNCS15	0.541	0.074	7.305	0.000
LNCS3	0.656	0.062	10.605	0.000
LNCS6	0.688	0.057	12.062	0.000
LNCS12	0.756	0.051	14.815	0.000
LNCS18	0.713	0.041	17.293	0.000

For factor score reliability

SAMPLE STATISTICS FOR ESTIMATED FACTOR SCORES

Means	VISION	VISION_SE
	0.000	0.194
Covariances		
	VISION	0.186

$$\rho = \frac{.186}{.186 + .194^2} = .832$$

Normalized Residuals for Covariances/Correlations/Residual Correlations

	LNCS15	LNCS3	LNCS6	LNCS12	LNCS18
LNCS15	0.000				
LNCS3	1.651	0.000			
LNCS6	-0.045	0.261	0.000		
LNCS12	-0.455	-0.241	0.021	0.000	
LNCS18	-0.629	-0.458	-0.177	0.353	0.000
FAR	-0.471	-0.731	-0.062	0.198	0.558

Local fit looks good as well...

Now we are ready for Step 2 for the Visual Impairment factor:
Generate plausible values using BAYES estimation of *previous MLR model*

ANALYSIS: ESTIMATOR = BAYES;

MODEL: ! Parameters from previously estimated measurement model, but all FIXED@

```
Vision BY far@1;
Vision BY lnCS15@0.49665(12);
Vision BY lnCS3@0.59434(13);
Vision BY lnCS6@0.76361(14);
Vision BY lnCS12@1.29636(15);
Vision BY lnCS18@1.50436(16);
```

```
[lnCS15@-3.69842];
[lnCS3@-3.93821];
[lnCS6@-3.72997];
[lnCS12@-2.36777];
[lnCS18@-1.40608];
[far@3.02632];
[vision@0];
```

```
lnCS15@0.13296(e2);
lnCS3@0.10479(e3);
lnCS6@0.14501(e4);
lnCS12@0.28191(e5);
lnCS18@0.48808(e6);
far@0.46002(e1);
vision@0.22350;
```

```
DATA IMPUTATION:           ! Creating plausible values for factor score
  NDATASETS = 100;         ! Number of separate values to create
  SAVE = Vision*.dat;      ! Name of separate datasets with plausible values
```

```
SAVEDATA:  FILE = VisionSummary.dat;  ! Summary about plausibles per person
           SAVE = FSCORES (100);      ! Needed to generate 100 factor scores
           FACTORS = Vision;          ! Which factors to save
```

```
Save file
  Vision*.dat
```

Order of variables

```
LNCS15
LNCS3
LNCS6
LNCS12
LNCS18
FAR
PARTID
VISION
```

Now we have 100 datasets (named Vision1.dat to Vision100.dat) with these variables in this order. Thus, rather than just using the mean of a person's factor score distribution, we are *sampling* from each person's factor distribution.

In addition, it made a text file called "Visionlist.dat" that lists these individual data files:

```
Vision1.dat
Vision2.dat
Vision3.dat
Vision4.dat
Vision5.dat
....
```

Later we will use these types of files to tell Mplus to run the same analysis on every single file, then aggregation the results (as in multiple imputation for missing data).

Measurement Model for Driving Impairment (including Omega)**Step 1: Estimate model using MLR, generate SVALUES text**

```

VARIABLE: ! Every variable in THIS MODEL
          USEVARIABLES = lane da_task crash stop speed time;
ANALYSIS: ESTIMATOR = MLR;
MODEL:    ! Measurement models
  Driving BY crash@1 ! 1 marker loading
          da_task* lane* stop* speed* time* (L2-L6);
  [lane* da_task* crash* stop* speed* time*]; ! All intercepts
  lane* da_task* crash* stop* speed* time* (E1-E6); ! Residual variances
  Driving*; [Driving@0]; ! Factor M=0, Var=?
  speed WITH time* (ResCov); ! Residual covariance

```

```

MODEL CONSTRAINT: ! TO GET OMEGA
NEW(SumLoad2 SumError SumRCov Omega);
SumLoad2 = (1+L2+L3+L4+L5+L6)^2;
SumError = E1+E2+E3+E4+E5+E6;
SumRCov = 2*(ResCov);
Omega = SumLoad2 / (SumLoad2+SumError+SumRCov);

```

MODEL FIT INFORMATION

Number of Free Parameters	19
Loglikelihood	
H0 Value	-37.119
H0 Scaling Correction Factor for MLR	1.1566
H1 Value	-30.710
H1 Scaling Correction Factor for MLR	1.1108

Information Criteria

Akaike (AIC)	112.239
Bayesian (BIC)	167.012
Sample-Size Adjusted BIC	106.915
(n* = (n + 2) / 24)	

Chi-Square Test of Model Fit

Value	12.791*
Degrees of Freedom	8
P-Value	0.1192
Scaling Correction Factor for MLR	1.0021

* The chi-square value for MLM, MLMV, MLR, ULSMV, WLSM and WLSMV cannot be used for chi-square difference testing in the regular way. MLM, MLR and WLSM chi-square difference testing is described on the Mplus website. MLMV, WLSMV, and ULSMV difference testing is done using the DIFFTEST option.

RMSEA (Root Mean Square Error Of Approximation)

Estimate	0.067
90 Percent C.I.	0.000 0.133
Probability RMSEA <= .05	0.293

CFI/TLI

CFI	0.922
TLI	0.854

Chi-Square Test of Model Fit for the Baseline Model

Value	76.677
Degrees of Freedom	15
P-Value	0.0000

SRMR (Standardized Root Mean Square Residual)

Value	0.054
-------	-------

Measurement Model Results from Step 1 for Driving:

MODEL RESULTS

	Estimate	S.E.	Est./S.E.	Two-Tailed P-Value
DRIVING BY				
CRASH	1.000	0.000	999.000	999.000
LANE	0.150	0.057	2.608	0.009
DA_TASK	0.173	0.074	2.348	0.019
STOP	0.347	0.163	2.124	0.034
SPEED	0.422	0.138	3.054	0.002
TIME	0.048	0.043	1.104	0.270
SPEED WITH				
TIME	-0.023	0.004	-5.393	0.000
Means				
DRIVING	0.000	0.000	999.000	999.000
Intercepts				
LANE	0.815	0.015	53.293	0.000
DA_TASK	0.256	0.013	20.102	0.000
CRASH	0.859	0.053	16.292	0.000
STOP	0.205	0.038	5.349	0.000
SPEED	0.836	0.042	19.687	0.000
TIME	3.146	0.009	349.081	0.000
Variances				
DRIVING	0.159	0.062	2.574	0.010
Residual Variances				
LANE	0.027	0.004	6.596	0.000
DA_TASK	0.017	0.004	4.613	0.000
CRASH	0.209	0.055	3.781	0.000
STOP	0.174	0.031	5.575	0.000
SPEED	0.210	0.028	7.391	0.000
TIME	0.010	0.001	8.639	0.000
New/Additional Parameters				
SUMLOAD2	4.578	1.185	3.865	0.000
SUMERROR	0.647	0.067	9.627	0.000
SUMRCOV	-0.046	0.009	-5.393	0.000
OMEGA	0.884	0.022	40.166	0.000

The SVALUES output option provided this code, which we will use to generate plausible values for our factor scores for later.

MODEL COMMAND WITH FINAL ESTIMATES USED AS STARTING VALUES

```
driving BY crash@1;
driving BY lane*0.14977 (12);
driving BY da_task*0.17282 (13);
driving BY stop*0.34713 (14);
driving BY speed*0.42198 (15);
driving BY time*0.04799 (16);
```

```
speed WITH time*-0.02305
(rescov);
```

```
[ lane*0.81538 ];
[ da_task*0.25614 ];
[ crash*0.85947 ];
[ stop*0.20455 ];
[ speed*0.83636 ];
[ time*3.14598 ];
[ driving*0 ];
```

```
lane*0.02734 (e1);
da_task*0.01669 (e2);
crash*0.20856 (e3);
stop*0.17387 (e4);
speed*0.20994 (e5);
time*0.01036 (e6);
driving*0.15881;
```

STANDARDIZED MODEL RESULTS
STDYX Standardization

	Estimate	S.E.	Est./S.E.	Two-Tailed P-Value
DRIVING BY				
CRASH	0.657	0.117	5.596	0.000
LANE	0.340	0.123	2.767	0.006
DA_TASK	0.470	0.132	3.576	0.000
STOP	0.315	0.115	2.748	0.006
SPEED	0.345	0.107	3.226	0.001
TIME	0.185	0.145	1.275	0.202
SPEED WITH				
TIME	-0.494	0.090	-5.478	0.000

For factor score reliability

SAMPLE STATISTICS FOR ESTIMATED FACTOR SCORES

Means	DRIVING	DRIVING_SE
	0.000	0.247

Covariances	DRIVING	DRIVING_SE
		0.098

$$\rho = \frac{.098}{.098 + .247^2} = .616$$

Normalized Residuals for Covariances/Correlations/Residual Correlations

	LANE	DA_TASK	CRASH	STOP	SPEED
LANE	0.000				
DA_TASK	-0.487	0.000			
CRASH	0.359	-0.390	0.000		
STOP	0.769	0.503	-0.004	0.000	
SPEED	0.458	-0.836	0.471	-0.482	0.000
TIME	-1.508	2.067	-0.346	-0.545	0.000

Local fit looks mostly ok...

Now we are ready for Step 2 for the Driving Impairment factor:
Generate plausible values using BAYES estimation of *previous MLR model*

ANALYSIS: ESTIMATOR = BAYES;

MODEL: ! Parameters from previously estimated measurement model, but all FIXED

```
driving BY crash@1;
driving BY lane@0.14977 (12);
driving BY da_task@0.17282 (13);
driving BY stop@0.34713 (14);
driving BY speed@0.42198 (15);
driving BY time@0.04799 (16);

speed WITH time@-0.02305 (rescov);
```

```
[ lane@0.81538 ];
[ da_task@0.25614 ];
[ crash@0.85947 ];
[ stop@0.20455 ];
[ speed@0.83636 ];
[ time@3.14598 ];
[ driving@0 ];
```

```
lane@0.02734 (e1);
da_task@0.01669 (e2);
crash@0.20856 (e3);
stop@0.17387 (e4);
speed@0.20994 (e5);
time@0.01036 (e6);
driving@0.15881;
```

DATA IMPUTATION: ! Creating plausible values for factor score
 NDATASETS = 100; ! Number of separate values to create
 SAVE = Driving*.dat; ! Name of separate datasets with plausible values

SAVEDATA: FILE = DrivingSummary.dat; ! Summary about plausibles per person
 SAVE = FSCORES (100); ! Needed to generate 100 factor scores
 FACTORS = Driving; ! Which factors to save

Save file
 Driving*.dat

Order of variables

```
LANE
DA_TASK
CRASH
STOP
SPEED
TIME
PARTID
DRIVING
```

Now we have 100 datasets (named Driving1.dat to Driving100.dat) with these variables in this order. Thus, rather than just using the mean, we are *sampling* from each person's factor score distribution.

In addition, it made a text file called "Visionlist.dat" that lists these individual data files:

```
Driving1.dat
Driving2.dat
Driving3.dat
Driving4.dat
Driving5.dat
....
```

Later we will use these types of files to tell Mplus to run the same analysis on every single file, then aggregation the results (as in multiple imputation).

Measurement Model for Attentional Impairment (including Omega)

Step 1: Estimate model using MLR, generate SVALUES text

```
VARIABLE: ! Every variable in THIS MODEL
          USEVARIABLES = lnda lnsa dscan;
ANALYSIS: ESTIMATOR = MLR;

MODEL:    ! Measurement models
          Attn BY lnda@1 ! 1 marker loading
              lnsa* dscan* (L2-L3);
          [lnda* lnsa* dscan*]; ! All intercepts
          lnda* lnsa* dscan* (E1-E3); ! Residual variances
          Attn*; [Attn@0]; ! Factor M=0, Var=?

MODEL CONSTRAINT: ! TO GET OMEGA
NEW(SumLoad2 SumError SumRCov Omega);
SumLoad2 = (1+L2+L3)^2;
SumError = E1+E2+E3;
SumRCov = 2*(0);
Omega = SumLoad2 / (SumLoad2+SumError+SumRCov);
```

The SVALUES output option provided this code, which we will use to generate plausible values for our factor scores for later.

```
MODEL COMMAND WITH FINAL ESTIMATES USED AS STARTING VALUES

attn BY lnda@1;
attn BY lnsa*0.51567 (l2);
attn BY dscan*1.10655 (l3);

[ lnda*4.35396 ];
[ lnsa*5.58076 ];
[ dscan*-0.01244 ];
[ attn*0 ];

lnda*0.51556 (e1);
lnsa*0.08112 (e2);
dscan*0.44855 (e3);
attn*0.44310;
```

Can you guess why I didn't include the model fit?

Measurement Model Results from Step 1 for Attention:

MODEL RESULTS

ATTN	BY	Estimate	S.E.	Est./S.E.	Two-Tailed P-Value
	LNDA	1.000	0.000	999.000	999.000
	LNSA	0.516	0.071	7.275	0.000
	DSCAN	1.107	0.139	7.933	0.000
Means					
	ATTN	0.000	0.000	999.000	999.000
Intercepts					
	LNDA	4.354	0.079	54.825	0.000
	LNSA	5.581	0.036	154.256	0.000
	DSCAN	-0.012	0.081	-0.154	0.878
Variances					
	ATTN	0.443	0.088	5.008	0.000
Residual Variances					
	LNDA	0.516	0.068	7.597	0.000
	LNSA	0.081	0.017	4.674	0.000
	DSCAN	0.449	0.086	5.243	0.000
New/Additional Parameters					
	SUMLOAD2	6.876	0.960	7.165	0.000
	SUMERROR	1.045	0.102	10.212	0.000
	SUMRCOV	0.000	0.000	0.000	1.000
	OMEGA	0.868	0.017	50.812	0.000

STANDARDIZED MODEL RESULTS

STDYX Standardization

ATTN	BY	Estimate	S.E.	Est./S.E.	Two-Tailed P-Value
	LNDA	0.680	0.055	12.275	0.000
	LNSA	0.770	0.055	14.087	0.000
	DSCAN	0.740	0.056	13.153	0.000

For factor score reliability

SAMPLE STATISTICS FOR ESTIMATED FACTOR SCORES

Means	
ATTN	ATTN_SE
0.000	0.313
Covariances	
ATTN	0.345

$$\rho = \frac{.345}{.345 + .313^2} = .779$$

Now we are ready for Step 2 for the Attentional Impairment factor:
Generate plausible values using BAYES estimation of *previous MLR model*

ANALYSIS: ESTIMATOR = BAYES;

MODEL: ! Parameters from previously estimated measurement model, but all FIXED

```
attn BY lnda@1;
attn BY lnsa@0.51567 (12);
attn BY dscan@1.10655 (13);
```

```
[ lnda@4.35396 ];
[ lnsa@5.58076 ];
[ dscan@-0.01244 ];
[ attn@0 ];
```

```
lnda@0.51556 (e1);
lnsa@0.08112 (e2);
dscan@0.44855 (e3);
attn@0.44310;
```

DATA IMPUTATION: ! Creating plausible values for factor score
 NDATASETS = 100; ! Number of separate values to create
 SAVE = Attn*.dat; ! Name of separate datasets with plausible values

SAVEDATA: FILE = AttnSummary.dat; ! Summary about plausibles per person
 SAVE = FSCORES (100); ! Needed to generate 100 factor scores
 FACTORS = Attn; ! Which factors to save

Save file
 Attn*.dat

Order of variables

```
LNDA
LNSA
DSCAN
PARTID
ATTN
```

Now we have 100 datasets (named Attn1.dat to Attn100.dat) with these variables in this order. Thus, rather than just using the mean of a person's factor score distribution, we are *sampling* from it.

In addition, it made a text file called "Attnlist.dat" that lists these individual data files:


```
Attn1.dat
Attn2.dat
Attn3.dat
Attn4.dat
Attn5.dat
....
```

Later we will use these types of files to tell Mplus to run the same analysis on every single file, then aggregation the results (as in multiple imputation).

Now we are ready to test the model of interest, as shown below (drawn by Mplus, made prettier by me). We'll begin with a saturated structural model that has main effects of the latent variables only.

VARIABLE: ! Every variable in THIS MODEL
 USEVARIABLES = lncs15 lncs3 lncs6 lncs12 lncs18 far
 lane da_task crash stop speed time
 lnda lnsa Dscan age75 lnps;

ANALYSIS: ESTIMATOR = MLR;



```

MODEL:      ! Measurement models
Vision BY far@1 lncs15* lncs3* lncs6* lncs12* lncs18*; ! 1 marker loading
[far* lncs15* lncs3* lncs6* lncs12* lncs18*]; ! All intercepts
far* lncs15* lncs3* lncs6* lncs12* lncs18*; ! Residual variances
Vision*; [Vision@0]; ! Factor M=0, Var=?

Driving BY crash@1 da_task* lane* stop* speed* time*; ! 1 marker loading
[lane* da_task* crash* stop* speed* time*]; ! All intercepts
lane* da_task* crash* stop* speed* time*; ! Residual variances
Driving*; [Driving@0]; ! Factor M=0, Var=?
speed WITH time* (ResCov); ! Residual covariance

Attn BY lnda@1 lnsa* dscan*; ! 1 marker loading
[lnda* lnsa* dscan*]; ! All intercepts
lnda* lnsa* dscan*; ! Residual variances
Driving*; [Driving@0]; ! Factor M=0, Var=?

Pspeed BY lns@1; lns@0; ! Bring proc speed into likelihood
[lns* Pspeed@0]; Pspeed*; ! Move its variance to factor

! Structural model with all possible main effects
Driving Vision Attn Pspeed ON Age75* (Age1-Age4); ! Age --> outcomes
Driving Pspeed Attn ON Vision* (Vis1-Vis3); ! Vision --> outcomes
Attn WITH Pspeed*; ! Res cov for Attn and Pspeed
Driving ON Pspeed* Attn* (Speed1 Attn1); ! Pspeed, Attn --> Driving
    
```

```

MODEL CONSTRAINT:
NEW(AgeVis AgeSpeed AgeAttn);
AgeVis = Age1*Vis1; ! Indirect effect of age to vision to driving
AgeSpeed = Age1*Speed1; ! Indirect effect of age to proc speed to driving
AgeAttn = Age1*Attn1; ! Indirect effect of age to attention to driving
    
```

MODEL FIT INFORMATION

Number of Free Parameters	58
Loglikelihood	
H0 Value	-1310.811
H0 Scaling Correction Factor for MLR	1.1063
H1 Value	-1238.221
H1 Scaling Correction Factor for MLR	1.0405

Information Criteria

Akaike (AIC)	2737.622
Bayesian (BIC)	2913.007
Sample-Size Adjusted BIC	2729.438
(n* = (n + 2) / 24)	

Chi-Square Test of Model Fit

Value	144.331*
Degrees of Freedom	110
P-Value	0.0156
Scaling Correction Factor for MLR	1.0059

* The chi-square value for MLM, MLMV, MLR, ULSMV, WLSM and WLSMV cannot be used for chi-square difference testing in the regular way. MLM, MLR and WLSM chi-square difference testing is described on the Mplus website. MLMV, WLSMV, and ULSMV difference testing is done using the DIFFTEST option.

RMSEA (Root Mean Square Error Of Approximation)

Estimate	0.045	
90 Percent C.I.	0.021	0.064
Probability RMSEA <= .05	0.635	

CFI/TLI

CFI	0.936
TLI	0.921

Chi-Square Test of Model Fit for the Baseline Model

Value	671.031
Degrees of Freedom	136
P-Value	0.0000

SRMR (Standardized Root Mean Square Residual)

Value	0.063
-------	-------

MODEL RESULTS

		Estimate	S.E.	Est./S.E.	Two-Tailed P-Value
VISION	BY				
	FAR	1.000	0.000	999.000	999.000
	LNCS15	0.481	0.099	4.837	0.000
	LNCS3	0.584	0.115	5.076	0.000
	LNCS6	0.759	0.136	5.583	0.000
	LNCS12	1.265	0.203	6.248	0.000
	LNCS18	1.491	0.232	6.416	0.000
DRIVING	BY				
	CRASH	1.000	0.000	999.000	999.000
	LANE	0.161	0.066	2.444	0.015
	DA_TASK	0.197	0.065	3.022	0.003
	STOP	0.381	0.164	2.330	0.020
	SPEED	0.418	0.164	2.540	0.011
	TIME	0.097	0.053	1.819	0.069
ATTN	BY				
	LNDA	1.000	0.000	999.000	999.000
	LNSA	0.491	0.061	8.000	0.000
	DSCAN	1.192	0.170	7.022	0.000
PSPEED	BY				
	LNPS	1.000	0.000	999.000	999.000

DRIVING ON				
VISION	-0.089	0.109	-0.814	0.415
PSPEED	0.114	0.083	1.387	0.165
ATTN	0.365	0.127	2.884	0.004
PSPEED ON				
VISION	0.167	0.100	1.658	0.097
ATTN ON				
VISION	0.287	0.137	2.095	0.036
DRIVING ON				
AGE75	0.001	0.011	0.119	0.905
VISION ON				
AGE75	0.024	0.011	2.187	0.029
ATTN ON				
AGE75	0.059	0.014	4.393	0.000
PSPEED ON				
AGE75	0.008	0.008	0.988	0.323
ATTN WITH				
PSPEED	0.061	0.027	2.292	0.022
SPEED WITH				
TIME	-0.025	0.004	-5.512	0.000
New/Additional Parameters				
AGEVIS	0.000	0.000	-0.119	0.905
AGESPEED	0.000	0.000	0.119	0.905
AGEATTN	0.000	0.000	0.123	0.902

STANDARDIZED MODEL RESULTS

STDYX Standardization

	Estimate	S.E.	Est./S.E.	Two-Tailed P-Value
VISION BY				
FAR	0.580	0.062	9.424	0.000
LNCS15	0.531	0.076	6.999	0.000
LNCS3	0.653	0.061	10.646	0.000
LNCS6	0.694	0.059	11.851	0.000
LNCS12	0.749	0.051	14.647	0.000
LNCS18	0.717	0.042	17.024	0.000
DRIVING BY				
CRASH	0.575	0.107	5.378	0.000
LANE	0.319	0.130	2.446	0.014
DA_TASK	0.470	0.100	4.694	0.000
STOP	0.302	0.115	2.630	0.009
SPEED	0.298	0.102	2.911	0.004
TIME	0.325	0.132	2.470	0.014
ATTN BY				
LNDA	0.672	0.058	11.501	0.000
LNSA	0.724	0.053	13.543	0.000
DSCAN	0.787	0.045	17.608	0.000
PSPEED BY				
LNPS	1.000	0.000	999.000	999.000
DRIVING ON				
VISION	-0.122	0.148	-0.826	0.409
PSPEED	0.164	0.120	1.368	0.171
ATTN	0.690	0.149	4.617	0.000
PSPEED ON				
VISION	0.160	0.094	1.715	0.086
ATTN ON				
VISION	0.209	0.096	2.191	0.028
DRIVING ON				
AGE75	0.017	0.148	0.118	0.906
VISION ON				
AGE75	0.224	0.087	2.582	0.010
ATTN ON				
AGE75	0.413	0.081	5.085	0.000
PSPEED ON				
AGE75	0.074	0.075	0.986	0.324
ATTN WITH				
PSPEED	0.221	0.088	2.523	0.012

```

SPEED WITH
TIME -0.544 0.090 -6.061 0.000
R-SQUARE
Latent Variable Estimate S.E. Est./S.E. Two-Tailed P-Value
VISION 0.050 0.039 1.291 0.197
DRIVING 0.532 0.151 3.526 0.000
ATTN 0.253 0.077 3.264 0.001
PSPEED 0.037 0.032 1.129 0.259

```

! Reduced structural model (no age or vision --> driving)

```

Vision Attn Pspeed ON Age75* (Age1-Age3); ! Age --> outcomes
Pspeed Attn ON Vision* (Vis1-Vis2); ! Vision --> outcomes
Attn WITH Pspeed*; ! Res cov for Attn and Pspeed
Driving ON Pspeed* Attn* (Speed1 Attn1); ! Pspeed, Attn --> Driving

```

MODEL FIT INFORMATION

```

Number of Free Parameters 56
Loglikelihood
H0 Value -1311.286
H0 Scaling Correction Factor 1.0933
for MLR
H1 Value -1238.221
H1 Scaling Correction Factor 1.0405
for MLR
Information Criteria
Akaike (AIC) 2734.572
Bayesian (BIC) 2903.909
Sample-Size Adjusted BIC 2726.670
(n* = (n + 2) / 24)
Chi-Square Test of Model Fit
Value 144.090*
Degrees of Freedom 112
P-Value 0.0221
Scaling Correction Factor 1.0142
for MLR
RMSEA (Root Mean Square Error Of Approximation)
Estimate 0.043
90 Percent C.I. 0.018 0.063
Probability RMSEA <= .05 0.691
CFI/TLI
CFI 0.940
TLI 0.927
SRMR (Standardized Root Mean Square Residual)
Value 0.063

```

Did constraining these two paths to 0 make the model worse?

Rescaled $-2\Delta LL(2) = 0.646$, $p = .72$, so no

This is the appropriate way to test a structural model, whose job is to reproduce the covariance among the latent factors and any observed predictors (but not among any observed predictors themselves).

Relying on good global model fit (which will mostly reflect the measurement models) is not sufficient to say a structural model fits.

What if we wanted to test an interaction between latent variables?

ANALYSIS: ESTIMATOR = MLR; TYPE = RANDOM; ALGORITHM = INTEGRATION;

SAVEDATA: SAVE = FSCORES; FILE IS FactorScores.dat; ! Save mean of plausibles only

! Reduced structural model (no age --> driving, but vision --> driving)

```

Vision Attn Pspeed ON Age75* (Age1-Age3); ! Age --> outcomes
Driving Pspeed Attn ON Vision* (Vis1-Vis3); ! Vision --> outcomes
Attn WITH Pspeed*; ! Res cov for Attn and Pspeed
Driving ON Pspeed* Attn* (Speed1 Attn1); ! Pspeed, Attn --> Driving

```

! Interaction between two latent variables (would be same if one was observed)

```

VisAttn | Vision XWITH Attn; ! VisAttn = new latent interaction
Driving ON VisAttn* (VxA); ! Interaction --> Driving

```

MODEL CONSTRAINT: ! latent factor variance of attn = .443, of vision = .224

NEW (V4low V4high A4low A4high);

V4low = Vis1 - VxA*SQRT(.443); ! Vision slope for -1SD attn

V4high = Vis1 + VxA*SQRT(.443); ! Vision slope for +1SD attn

A4low = Attn1 - VxA*SQRT(.224); ! Attn slope for -1SD vision

A4high = Attn1 + VxA*SQRT(.224); ! Attn slope for +1SD vision

```

MODEL FIT INFORMATION
Number of Free Parameters          58
Loglikelihood
  H0 Value                        -1310.275
  H0 Scaling Correction Factor     1.1008
  for MLR
Information Criteria
  Akaike (AIC)                    2736.549
  Bayesian (BIC)                  2911.934
  Sample-Size Adjusted BIC        2728.365
  (n* = (n + 2) / 24)

```

Model fit has disappeared once we've used numeric integration (no covariance matrix to come back to anymore). STDYX disappears for the same reason.

New structural model output only:

```

DRIVING    ON
  VISION    -0.105    0.116    -0.907    0.365 simple vision slope at attn=0
  PSPEED    0.117    0.084    1.384    0.166
  ATTN      0.372    0.088    4.204    0.000 simple attn slope at vision=0
  VISATTN   0.139    0.143    0.957    0.330
PSPEED     ON
  VISION    0.168    0.101    1.663    0.096
ATTN       ON
  VISION    0.304    0.143    2.128    0.033
VISION     ON
  AGE75     0.024    0.011    2.189    0.029
ATTN       ON
  AGE75     0.060    0.013    4.453    0.000
PSPEED     ON
  AGE75     0.008    0.008    0.983    0.326
ATTN       WITH
  PSPEED    0.060    0.027    2.232    0.026
New/Additional Parameters
  V4LOW     -0.198    0.170    -1.166    0.244 simple vision slope at attn=-1SD
  V4HIGH    -0.102    0.127    -0.097    0.923 simple vision slope at attn=+1SD
  A4LOW     0.306    0.105    2.908    0.004 simple attn slope at vision=-1SD
  A4HIGH    0.438    0.117    3.738    0.000 simple attn slope at vision=+1SD

```

What if we wanted to test multiple interactions between latent variables, or the model wouldn't converge (or there are too many latent variables to estimate all at once)? **Plausible values to the rescue!** So far we've done step 1 (run model in MLR) and step 2 (use Bayes estimation to generate plausible values for each latent factor). Now we need steps 3 and 4.

Step 3: Merge the plausible values of different factors together, so that you have 100 complete datasets. Here is a SAS macro to automate that process:

```

*****
****          INFO NEEDED TO ENTER TO MERGE PLAUSIBLE VALUES          ****
*****

* Folder for plausible files;      %LET filesave = C:\Dropbox\14_Psyc948\DriverScan\PV Files;
* SAS file name of original data; %LET datafile = MyOriginalData;
* Name of person ID variable;     %LET IDvar = PartID;
* Suffix # of FIRST file;         %LET startP = 1;
* Suffix # of LAST file;          %LET endP = 100;
* Total # of sets of files;       %LET total = 3;

%MACRO LabelThem;
* index = count of how many sets of files,
* prefix = name of file prefix;
* # items needed to drop from front of file;
%IF &index. = 1 %THEN %DO; %LET prefix = Vision; %LET ndrop = 6; %END;
%IF &index. = 2 %THEN %DO; %LET prefix = Driving; %LET ndrop = 6; %END;
%IF &index. = 3 %THEN %DO; %LET prefix = Attn; %LET ndrop = 3; %END;
**** REPEAT THIS FOR ALL YOUR SETS OF FILES TO BE MERGED ****
%MEND LabelThem;

```

```

*****;
****
      NOTHING NEEDS TO BE CHANGED FROM HERE, JUST RUN IT
*****;
*****;
* Sort original data by ID; PROC SORT DATA=&datafile.; BY &IDvar.; RUN;
%GLOBAL index prefix ndrop; %MACRO Import;
%DO num=&startp. %TO &endp.; DATA Merge&num.; SET &datafile.; RUN;
  %DO index=1 %TO &total.; %LabelThem; * Import plausible file;
    DATA &prefix.&num.; INFILE "&filesave.\&prefix.&num..dat" DLM=TAB LRECL=1000;
    LENGTH var1-var&ndrop. $3; * Items as string vars given * missing;
    INPUT var1-var&ndrop. &IDvar. &prefix.; KEEP &IDvar. &prefix.; RUN;
    PROC SORT DATA=&prefix.&num.; BY &IDvar.; RUN;
    * Merge with original data, replace missing values;
    DATA Merge&num.; MERGE Merge&num. &prefix.&num.; BY &IDvar.;
    IF &prefix.=. THEN &prefix.=-9999; IF &IDvar.=. THEN &IDvar.=-9999; RUN;
    * Remove SAS datasets;
    PROC DATASETS LIB=WORK NOLIST; DELETE &prefix.&num.; RUN; QUIT;
  %END; * Export to .csv for use in Mplus;
    PROC EXPORT DATA=Merge&num. OUTFILE= "&filesave.\PV&num..csv"
      DBMS=CSV REPLACE; PUTNAMES=NO; RUN;
    * Remove SAS datasets;
    PROC DATASETS LIB=WORK NOLIST; DELETE Merge&num.; RUN; QUIT;
%END; %MEND Import; * Run macro; %Import;
* Build list of plausible values files;
%MACRO Makelist;
DATA _NULL_; * Name of file to print to;
  FILE "&filesave.\PVFilesList.dat" NOPAD NOTITLES;
  * Print all dataset names;
  %DO i=&startP. %TO &endP.; PUT @1 "PV&i..csv"; %END;
RUN; %MEND Makelist; * Run macro; %Makelist;
*****;

```

When the SAS program is done running, you will have 100 .csv files with all latent variables merged together, as well as a file called "PVFileList.dat" that lists all these files. Now we are ready to analyze!

Step 4: Estimate the same model, but using the plausible values instead of the latent factors to build an observed interaction term. This tells Mplus to do so for all 100 files and then combine the results.

```

TITLE: SEM Example for Driverscan using Plausible Values;
DATA:
  FILE = PVFilesList.dat;      ! FILE that lists all the data file names
  TYPE = IMPUTATION;          ! Analyze and combine results across files
VARIABLE:
! List of ALL variables in data file
  NAMES = PartID sex age75 lncs15 lncs3 lncs6 lncs12 lncs18 far lnps
         lnda lnsa Dscan lane da_task crash stop speed time ticket acc
         Vision Driving Attn; ! New factor scores
! Variables to be analyzed in this model
  USEVARIABLE = age75 lnps Vision Driving Attn VisAttn;
! Missing data identifier
  MISSING ARE ALL (-9999);
! ID variable;
  IDVARIABLE IS PartID;
DEFINE:   VisAttn = Vision * Attn; ! Now interaction is observed variable
ANALYSIS: ESTIMATOR = MLR;
OUTPUT:   STDYX RESIDUAL;          ! Standardized model, local fit
          SAMPSTAT;                ! Get descriptive stats for variables
MODEL: ! Structural model
  Vision Attn lnps ON Age75* (Age1-Age3); ! Age --> outcomes
  Driving lnps Attn ON Vision* (Vis1-Vis3); ! Vision --> outcomes
  Attn WITH lnps*; ! Res cov for Attn and Pspeed
  lnps Attn Vision ON VisAttn*; ! Interaction relations
  Driving ON lnps* Attn* (Speed1 Attn1); ! Obs vars --> Driving
  Driving ON VisAttn* (VxA); ! Obs var interaction
MODEL CONSTRAINT: ! Plausible score variance of attn = .441, of vision = .221
NEW (V4low V4high A4low A4high);
V4low = Vis1 - VxA*SQRT(.441); ! Vision slope for -1SD attn
V4high = Vis1 + VxA*SQRT(.441); ! Vision slope for +1SD attn
A4low = Attn1 - VxA*SQRT(.221); ! Attn slope for -1SD vision
A4high = Attn1 + VxA*SQRT(.221); ! Attn slope for +1SD vision

```

```

Number of Free Parameters                21
Loglikelihood
  H0 Value
    Mean                                -391.439
    Std Dev                              9.112
  H1 Value
    Mean                                -390.891
    Std Dev                              9.052
Information Criteria
  Akaike (AIC)
    Mean                                824.879
    Std Dev                              18.224
  Bayesian (BIC)
    Mean                                888.380
    Std Dev                              18.224
  Sample-Size Adjusted BIC (n* = (n + 2) / 24)
    Mean                                821.916
    Std Dev                              18.224
Chi-Square Test of Model Fit
  Degrees of freedom                    1
  Mean                                  1.268
  Std Dev                               1.630
RMSEA (Root Mean Square Error Of Approximation)
  Mean                                  0.039
  Std Dev                               0.053
CFI/TLI
  CFI
    Mean                                  0.986
    Std Dev                              0.027
  TLI
    Mean                                  0.914
    Std Dev                              0.443
SRMR (Standardized Root Mean Square Residual)
  Mean                                  0.013
  Std Dev                               0.008

```

Each dataset has its own set of fit statistics, the mean and SD for each are given here (with much other output omitted).

This means that likelihood ratio testing cannot proceed in the usual way.

MODEL RESULTS

	Estimate	S.E.	Est./S.E.	Two-Tailed P-Value	Rate of Missing
VISION ON					
AGE75	0.019	0.009	2.137	0.033	0.153
VISATTN	0.082	0.251	0.325	0.745	0.255
ATTN ON					
AGE75	0.042	0.012	3.647	0.000	0.129
VISION	0.454	0.123	3.700	0.000	0.171
VISATTN	0.055	0.247	0.222	0.824	0.218
LNPS ON					
AGE75	0.009	0.008	1.155	0.248	0.014
VISION	0.140	0.093	1.502	0.133	0.120
VISATTN	-0.012	0.144	-0.086	0.931	0.255
DRIVING ON					
VISION	-0.060	0.102	-0.589	0.556	0.276
LNPS	0.095	0.079	1.201	0.230	0.208
ATTN	0.187	0.066	2.839	0.005	0.279
VISATTN	0.013	0.133	0.101	0.919	0.320
ATTN WITH					
LNPS	0.045	0.023	1.922	0.055	0.111
New/Additional Parameters					
V4LOW	-0.069	0.133	-0.518	0.605	0.306
V4HIGH	-0.051	0.136	-0.375	0.708	0.286
A4LOW	0.180	0.089	2.028	0.043	0.284
A4HIGH	0.193	0.092	2.087	0.037	0.312

What would have happened if we used the mean of each person's factor score distribution from the single-factor models as a single observed variable instead? Let's examine two versions.


```

TITLE: SEM Example for Driverscan using Factor Score Values;
DATA:
  FILE = SEMfactorscores.dat;      ! Mean factor score merged back into data
  TYPE = INDIVIDUAL;              ! Now just a regular analysis
VARIABLE:
  ! List of ALL variables in data file
  NAMES = PartID sex age75 lnps vision driving attn; ! New factor scores
  ! Variables to be analyzed in this model
  USEVARIABLE = age75 lnps vision driving attn visattn;
  ! Missing data identifier
  MISSING ARE ALL (-9999);
  ! ID variable;
  IDVARIABLE IS PartID;
DEFINE:    VisAttn = Vision * Attn; ! Now interaction is observed variable

ANALYSIS: ESTIMATOR = MLR;
OUTPUT:   STDYX RESIDUAL;      ! Standardized model, local fit
             SAMPSTAT;           ! Get descriptive stats for variables
MODEL:
  ! Structural model
  Vision Attn lnps ON Age75* (Age1-Age3); ! Age --> outcomes
  Driving lnps Attn ON Vision* (Vis1-Vis3); ! Vision --> outcomes
  Attn WITH lnps*;                       ! Res cov for Attn and Pspeed
  lnps Attn Vision ON VisAttn*;          ! Interaction relations
  Driving ON lnps* Attn* (Speed1 Attn1); ! Obs vars --> Driving
  Driving ON VisAttn* (VxA);             ! Obs var interaction

MODEL CONSTRAINT: ! New factor score variance of attn = .345, of vision = .186
NEW (V4low V4high A4low A4high);
V4low = Vis1 - VxA*SQRT(.345); ! Vision slope for -1SD attn
V4high = Vis1 + VxA*SQRT(.345); ! Vision slope for +1SD attn
A4low = Attn1 - VxA*SQRT(.186); ! Attn slope for -1SD vision
A4high = Attn1 + VxA*SQRT(.186); ! Attn slope for +1SD vision

MODEL FIT INFORMATION
Number of Free Parameters          21
Loglikelihood
  H0 Value                          -323.603
  H0 Scaling Correction Factor      1.1794
  for MLR
  H1 Value                          -323.451
  H1 Scaling Correction Factor      1.1676
  for MLR
Information Criteria
  Akaike (AIC)                      689.207
  Bayesian (BIC)                    752.708
  Sample-Size Adjusted BIC          686.244
  (n* = (n + 2) / 24)
Chi-Square Test of Model Fit
  Value                              0.331*
  Degrees of Freedom                 1
  P-Value                            0.5649
  Scaling Correction Factor          0.9182
  for MLR
* The chi-square value for MLM, MLMV, MLR, ULSMV, WLSM and WLSMV cannot be used
for chi-square difference testing in the regular way. MLM, MLR and WLSM
chi-square difference testing is described on the Mplus website. MLMV, WLSMV,
and ULSMV difference testing is done using the DIFFTEST option.

RMSEA (Root Mean Square Error Of Approximation)
  Estimate                          0.000
  90 Percent C.I.                   0.000 0.178
  Probability RMSEA <= .05          0.633
CFI/TLI
  CFI                               1.000
  TLI                               1.172
SRMR (Standardized Root Mean Square Residual)
  Value                              0.007

```

Here's what the code would have looked like to correct for factor score unreliability:

```

! Measurement models for factor score unreliability
Fvision BY Vision@1; Vision*(ResVis); [Fvision@0 Vision*];
Fattn BY Attn@1; Attn*(ResAttn); [Fattn@0 Attn*];
Fdrive BY Driving@1; Driving*(ResDriv); [Fdrive@0 Driving*];

! Structural model with "latent" variables;
Fvision Fattn lnps ON Age75* (Age1-Age3); ! Age --> outcomes
Fdrive lnps Fattn ON Fvision* (Vis1-Vis3); ! Vision --> outcomes
Fattn WITH lnps*; ! Res cov for Attn and Pspeed
lnps Fattn Fvision ON VisAttn*; ! Interaction relations
Fdrive ON lnps* Fattn* (Speed1 Attn1); ! Obs vars --> Driving
Fdrive ON VisAttn* (VxA); ! Obs var interaction

MODEL CONSTRAINT: ! New factor score variance of attn = .345, of vision = .186
NEW (V4low V4high A4low A4high);
V4low = Vis1 - VxA*SQRT(.345); ! Vision slope for -1SD attn
V4high = Vis1 + VxA*SQRT(.345); ! Vision slope for +1SD attn
A4low = Attn1 - VxA*SQRT(.186); ! Attn slope for -1SD vision
A4high = Attn1 + VxA*SQRT(.186); ! Attn slope for +1SD vision

ResVis =(1-.832)*0.186; ! Fix residual variances to "unreliable" part
ResAttn=(1-.779)*0.345;
ResDriv=(1-.616)*0.097;

```

MODEL RESULTS

	Estimate	S.E.	Est./S.E.	Two-Tailed P-Value	P-value from Rel	P-value from SEM	P-value from Plaus
VISION ON							
AGE75	0.019	0.007	2.563	0.010	.010	.029	.033
VISATTN	0.170	0.259	0.654	0.513	.513		.745
ATTN ON							
AGE75	0.047	0.010	4.632	0.000	.000	.000	.000
VISION	0.227	0.105	2.165	0.030	.034	.033	.000
VISATTN	0.092	0.240	0.385	0.700	.731		.824
LNPS ON							
AGE75	0.009	0.008	1.094	0.274	.320	.326	.248
VISION	0.168	0.095	1.767	0.077	.077	.096	.133
VISATTN	-0.032	0.146	-0.220	0.826	.796		.931
DRIVING ON							
VISION	-0.022	0.064	-0.343	0.731	.515	.365	.556
LNPS	0.080	0.052	1.538	0.124	.242	.166	.230
ATTN	0.229	0.040	5.791	0.000	.000	.000	.005
VISATTN	0.024	0.090	0.381	0.703	.774	.330	.919
ATTN WITH							
PSPEED	0.052	0.020	2.580	0.010	.012	.026	.055
New/Additional Parameters							
V4LOW	-0.042	0.084	-0.500	0.617	.511	.244	.605
V4HIGH	-0.002	0.082	-0.020	0.984	.689	.923	.708
A4LOW	0.214	0.057	3.789	0.000	.000	.004	.043
A4HIGH	0.244	0.055	4.475	0.000	.000	.000	.047

The p-values from the model treating factor scores as observed variables are compared here with those correcting for factor score unreliability, from SEM, and from using plausible values. As shown, results are similar, although relative to SEM, although slightly more conservative when using plausible values, and slightly more liberal when using factor scores. To the extent that reliability of the sum score or the factor scores is lower, though, these results should diverge more.